

抗生素效价测定中微型电子计算机的应用

洪 雨 文

(安徽师范大学生物系, 芜湖)

在抗生素管碟法效价测定中, 通常采用一剂量法。其中标准曲线制作过程往往需要回归, 涉及烦琐的计算。如果采用微型电子计算机, 编入简易的程序, 很快就能将各批实验数据代

入, 求得回归直线的截距和斜率, 并能立即根据样品组抑菌圈值求得样品浓度。

设抗生素标准液共 n 个浓度 (n 可根据情况自己选定), 分别为 $y_1, y_2, \dots, y_n$ 。通过实验得

出对应的抑菌圈直径校正值分别为 $x_1, x_2, \dots, x_n$ 。实验同时测出未知样品抑菌圈直径校正值为 X 。

抗生素的浓度 Y 与抑菌圈的直径 X 成曲线关系, 而抗生素浓度的对数值与抑菌圈的直径成直线关系。即 $\log Y = BX + A$

由于一般实验得到的坐标点往往不是很精确地落在一直线上, 因而必须进行直线回归。直线回归可有多种方法, 例如解方程组:

$$\begin{cases} \log y_1 = Bx_1 + A \\ \log y_2 = Bx_2 + A \\ \dots\dots\dots \\ \log y_n = Bx_n + A \end{cases}$$

求得截距 A 和斜率 B 。

还可根据公式计算, 方法都十分繁琐。

如果使用电子计算机, 可以采用以下程序:

```
10 PRINT "N = ";
20 INPUT N
30 LET X = 0
40 LET X1 = 0
50 LET Y = 0
60 LET Y1 = 0
70 LET Z = 0
80 FOR I = 1 TO N
90 READ C, D
100 LET X = X + C
110 LET X1 = X1 + C * C
120 LET Y = Y + log D
130 LET Y1 = Y1 + (log D) * (log D)
140 LET Z = Z + C * (log D)
150 NEXT I
160 LET L1 = X1 - X * X / N
170 LET L2 = Y1 - Y * Y / N
180 LET L3 = Z - X * Y / N
190 LET B = L3 / L1
200 LET A = (Y - B * X) / N
210 LET R = L3 / SQRT(L1 * L2)
220 PRINT "log Y = "; B; "X + (" ; A; ")"
230 PRINT "R = "; R
240 PRINT "IF X = ";
```

250 INPUT C

260 PRINT "log Y = "; B * C + A

270 PRINT "Y = "; 10 ↑ (B * C + A)

280 END

290 DATA $x_1, y_1, x_2, y_2, x_3, y_3, \dots, x_n, y_n$

该程序中 N 为抗生素标准液组数。 X, X_1 分别为自变量 X 的和与平方和。 Y, Y_1 分别为函数 Y 的和与平方和。 Z 为 X 与 Y 的乘积。 L_1, L_2 分别为 X, Y 的离差平方和。 L_3 为 X 与 Y 的离差乘积。 A 为截距。 B 为斜率。 R 为相关系数。 $x_1, \dots, x_n$ 为各组抑菌圈校正值。 $y_1, \dots, y_n$ 为各组标准液浓度。

按照该程序运算, 输入 N 及 $x_1, x_2, \dots, x_n; y_1, y_2, \dots, y_n$ 可以立即打印出回归后的直线方程式及其相关系数。接着再从键盘输入抗生素样品的抑菌圈直径校正值 C , 即可打印出该样品浓度的对数, 即 $\log Y$, 并换算出浓度值 Y 。

该程序采用基本 BASIC 语言编写, 适用于各种不同型号的电子计算机。

对于某些组装有回归分析计算机能的计算机, 例如 FX-702P, 程序可更加简化如下:

```
10 SAC
20 PRINT "log Y = B * X + A"
30 PRINT "N = ";
40 INPUT N
50 FOR I = 1 TO N
60 PRINT "X("; I; ") = ";
70 INPUT X
80 PRINT "Y("; I; ") = ";
90 INPUT Y
100 STAT X, log Y
110 NEXT I
120 PRINT "LRA = "; LRA
130 PRINT "LRB = "; LRB
140 PRINT "COR = "; COR
150 PRINT "X = ";
160 INPUT X
170 PRINT "log Y = "; EOYX
180 PRINT "Y = "; 10 ↑ EOYX
190 END
```

此程序中出现了一些该计算机使用的函数名称,如 LRA, LRB, COR 及 EOYX, 分别表示截距、斜率、相关系数及 Y 的估定值。

目前,电子计算机已在许多地方普及使用。采用以上简易可行的程序能省去大量人工计算,肯定是可取的。